MICROWAVE LINK PREDICTIONS

PROGRAM DESCRIPTION **RADLNK**

WRITTEN BY: D. BURGER JULY 1985

TABLE OF CONTENTS

1. PROGRAMME DESCRIPTION

1.1 INTRODUCTION

The program RADLNK is designed to provide the microwave radio system designer a consistant means of obtaining a realistic evaluation of a microwave radio link. The link performance output can accommodate a basic system with variations such a frequency/space diversity, space grid coordinates, flexible equipment selection etc. Refer to the sample printout attached.

A segment on Path Geometry is included in order for the designer to evaluate path obstructions and tower heights. This is a graphics-style package and affects the link performance with respect to RMS Roughness of the terrain. The Path Geometry segment also shows the radio rayline clearance from the ground.

Part of the Path Geometry block is a section on Ground/Sea reflection analysis. This scans the terrain for potential reflective regions and presents a report detailing these regions. The analysis also recommends a space diversity separation for both ends of the link.

The designer then may examine the reflective region on a site visit and decide if space diversity need by used.

PROGRAM RADLNK is suitable for use on microwave radio equipment of all types, in all terrain areas and all realistic K factors.

The program does have limitations and there are summarized below:

Frequency of operation	0.001 GHz (1MHz) to 150 GHz.
For rain considerations	2 GHz to 99 Ghz.
For atmospheric absorption	2 GHz to 99 GHz
Path Length.	0.001 km (1m) to 999.999 km.

The effect of the K factor on Earth Bulge is always considered. This permits the accurate profiling of a radio path with different K factors expected in regions.

The program is written in GWBASIC for the IBM PC and variants.

1.1.1 DESIGNING MICROWAVE RADIO LINKS

The starting point in a radio communications link is to identify it as an economic and practical means of bridging awkward terrain and flexible expansion of capacity in the future. The planning of a link must take into account, cost, equipment availability and propagation availability.

The designer may adopt "rules" on which to base a link or network. Such rules take the form of:

1. Minimization of cost, spares inventory, maintenance.
2. Propagation availability greater than 99.950% End to End.
3. Hardware availability greater than 99.900%.
4. Terrain clearance of rayline greater than 0.3F at K = 0.66 and F at K = 1.33, where F is Fresnel zone radius.

Program RADLNK can consider points 2) and 4) but it is up to the designer to specify the rules. Point 3) is the subject of another program titled "COMREL; Communications Network Reliability, Hardware Considerations. Point 1) is up to the ingenuity of the designer.

The designer may determine an equipment configuration to meet a specified requirement, or use "standard" equipment and obtain an availability figure to offer a client.

1.1.2 INFLUENCE OF TERRAIN

The microwave beam is influenced by the intermediate terrain between the stations and obstacles in its path. It follows a slightly curved path in the vertical plane, ie, it is refracted in the atmosphere due to the dielectric constant of the atmosphere. The amount of refraction changes with temperature, vapour pressure and relative humidity. In addition to this effect, the microwave beam can be diffracted by obstacles in it path or in the extreme, the scattering effect from the troposphere. It is up to the designer to evaluate any diffraction or troposcatter effects separately from free-space data used in RADLNK. If a path is surveyed using optical means the results would correspond to a K factor of approx 1.16. Obstacles such as trees should be considered as blocking a microwave beam, so adequate clearance should be considered even allowing for their growth.

The consideration of the Fresnel Zone clearance becomes paramount in highly reflective paths, eg over water. Rayline clearance to the 2nd Fresnel zone is important over this type of path.

1.1.3 INFLUENCE OF WEATHER

At microwave frequencies over 2 GHz, rain and preciption can influence the attenuation on a route. Refer to the specific attenuation chart elsewhere in this handbook. The rain attenuation model in program RADLNK is modelled from this information in this chart. The effect of atmospheric absorption is presented, where, at certain frequencies, molecular resonance occurs. This can be significant at frequencies near 21 GHz and 60 GHz which correspond to the water vapour and oxygen resonance frequencies respectively.

1.1.4 SITUATIONS TO BE AVOIDED

Several situations should be avoided when designing a link:

1. Over water paths and paths over low flat terrain. Where they cannot be avoided, the use of space diversity or the High-low placement of antennas to move the reflection point onto rough terrain is recommended.

2. Sites near radar stations. Possible solutions here include RF filters, RF attenuators and utilization of terrain protection, ie placing a terminal in a radar shadow region.

3. Sites on the very top of high mountains or ridges are suseptible to more potential interference from distant links, and the effects of delay distortion from multiple path effects.

4. Having a microwave rayline pass near city buildings or down streets, results in severe reflection situations and increased outage times.

1.2 SUBROUTINES (LINK BUDGET)

1.2.1 SYSTEM DETAILS

A brief description of each subroutine (Block) utilized in the Link Budget calculation follows:

Data Input Block	Inputs station names, path distance, grid bearings, diversity selection, K factor. The option for a link utilizing both space and frequency diversity cannot be directly supported. The program would need to be run twice with each option.

1.2.2 RAIN DATA

Data Input Block	Inputs rain rate data and climate zone. The climate zone is based on the geographic location of the link. N.B. Rain Data from Rain Contours 0.05% Probability elsewhere in this handbook.

1.2.3 EQUIPMENT

Data Input Block

Inputs type of equipment, allows selection of antenna size and peculiar equipment variations. The equipment shown in the menu is popular for links up to 8 Mb/s (120 channels) capacity. Any equipment may be configured providing specifications are known. Equipment without integral antennas would need some default antenna gains to be entered in the data block. Refer to equipment Data block.

1.2.4 PERFORMANCE REPORT OUTPUT

Calculation/Output Block

Calculation of path loss (free space) fade margin, Fading, 1st Fresnel zone radius and system availabilities. Where options like diversity are not selected, they do not appear on output. It is at the start of this block that all link calculations are performed.

1.2.5 EQUIPMENT DATA

Data Block

Hardware data not directly accessed from user. It is here that the user may enter other equipment as options.

1.2.6 AMPLITUDE DISPERSION FADING

Calculation Block — Fading calculation based on Algorithm from NEC DIGITAL MICROWAVE RADIO HANDBOOK for digital radio links. The fading prediction is applicable to FDM and video links.

1.2.7 RAIN FADING

Calculation Block — Fading calculation based on Algorithm derived from CCIR REP 721-1

1.2.8 STATION DISTANCES & BEARINGS

Calculation Block — Calculates distance between stations given grid coordinates and the bearing of station B from station A. This block operates using the spherical "Great Circle" calculations.

1.3 SUBROUTINE (PATH GEOMETRY)

A brief description of each subroutine (Block) utilized in the Path Geometry calculation follows:

1.3.1 PATH GEOMETRY

Menu Block — SUB-MENU SELECTION

1.3.2 SUPPORT HEIGHTS

Data Input Block — Inputs height of support structures at both stations. Radio Horizon distance shown for smooth earth. The radio horizon distance is presented as a course of station support heights, and suitable in an interference study.

1.3.3 PATH PROFILE ENTRY

Data input Block

Inputs path profile data from map. The number of input points is not critical.

1.3.4 PATH DATA RESET

Calculation Block

Resets path data to zero if a mistake or new data is inputted. Does not affect Link Budget subroutines.

1.3.5 DEFAULT VALUES SET

1.3.6 PATH DATA PARAMETER OUTPUT

Output Block

Dumps path profile data onto printer.

1.3.7 PROFILE DISPLAY

Menu Block

SUB-SUB MENU SELECTION.

1.3.8 DOT JOINING/DATA FILL

Calculation Block

Joins all the path profile data points (linear).

1.3.9 EARTH BULGE CORRECTION

Calculation Block

Modifies the data fill to account for earth Bulge.

1.3.10 RMS ROUGHNESS

Calculation Block

Calculates the RMS roughness of corrected data. This is used in LINK BUDGET. Default values are set.

1.3.11 RAYLINE CLEARANCE

Calculation Block

Calculates the clearance between the rayline and the terrain. This is titled CLR on the plotter output.

1.3.12 PLOT SCALING

Calculation Block — Determines maximum and minimum data and calculates suitable plot dimensions.

1.3.13 PLOTTER

Output Block — Plots data size A3 of A4 or any specified size.

1.4 GROUND/SEA REFLECTION ANALYSIS

A brief description of each subroutine (Block) utilized in the Reflection Analysis follows:

1.4.1 SELECTION AND TESTING REFLECTIONS — Scans through the region selected by the designer, or due to illumination or full path. (Note: Full path testing is time consuming). Testing performed at 3 K values.

1.4.2 ILLUMINATED REGION

Calculation Block — This scans the path profile to determine the illuminated region open for reflections.

1.5 PROGRAM VARIABLES

F1	-	FREQUENCY (GHZ)
P1	-	PATH LENGTH (KM)
P2	-	TRANSMIT POWER (DBM)
P9	-	FREE SPACE PATH LOSS (DB)
G1,G2	-	ANTENNA GAIN (DBI)
G3	-	OTHER ATTENUATION (DB)
F8	-	RAIN MARGIN
F9	-	FADE MARGIN (DB)
R1	-	RECEIVE THRESHOLD (DBM)
R2	-	RAIN RATE EXCEEDED R4% PER MONTH
R3	-	CLIMATE FACTOR
R4	-	RAINFALL RATE PERCENTILE VALUE
R9	-	RECEIVE LEVEL (DBM)
A8	-	RAIN ATTENUATION
PRM	-	RAIN OUTAGE TIME PER MONTH
PRY	-	RAIN OUTAGE TIME PER YEAR
O1	-	AMPLITUDE DISPERSION OUTAGE TIME
O2	-	DIVERSITY LINK OUTAGE TIME
O8	-	SYSTEM AVAILABILITY PER YEAR
O9	-	SYSTEM AVAILABILITY PER MONTH
D1,D2	-	SUPPORT HEIGHTS FOR STATIONS
N1$,		
N2$	-	STATION NAMES
N3$	-	EQUIPMENT NAMES
QX$	-	INPUT VARIABLE FOR MENU'S
Z $	-	DATA DISPLAY FIELD
AA1,AA2	-	COORDINATES OF STATION N1$
BB1,BB2	-	COORDINATES OF STATION N2$
H	-	BEARING OF STATION N2$ FROM STATION N1$
XD(101),		
YD(101)	-	MATRIX WHERE INPUT PATH DATA IS ENTERED
X(101),		
Y(101)	-	PLOTTER DATA
RF(101)	-	SURFACE ROUGHNESS CALCULATION DATA
U5	-	NUMBER OF X VALUES IN PLOT

U6	-	NUMBER OF Y VALUES IN PLOT
R	-	PLOTTER X AXIS INCREMENT
L	-	PLOTTER Y AXIS INCREMENT
KN	-	COUNTER FOR DATA X VALUES CLOSER THAN 0.5R
KF	-	K FACTOR
Y1, Y2	-	MINIMUM, MAXIMUM PLOT VALUES
KK	-	NUMBER ROUNDING FACTOR
T44	-	EARTH BULGE (MAXIMUM POINT)
T46	-	RMS ROUGHNESS
T48	-	FRESNEL ZONE RADIUS (MAXIMUM, MID PATH)
SD1	-	SPACE DIVERSITY SEPARATION (M)
FD1	-	FREQUENCY DIVERSITY SEPARATION (MHZ)
ALF	-	ANGULAR VARIABLE FOR REFLECTION CALCULATION
BET	-	ANGULAR VARIABLE FOR REFLECTION CALCULATION
MI$	-	Metric / Imperial units selection.
ME$	-	User message / name

2. PROGRAM OPERATION

2.1 DESCRIPTION

Program RADLNK effectively consists of three separate programs with the only interlinked variables P1, KF and T46. The three programs consist of the Link Budget Block, Path Geometry Block, and Ground/sea reflection analysis. All are menu driven for ease of use. The output procedure follows the following steps, for full variable interlinking.

a) Enter all data, no particular order
b) Print Path Data (optional)
c) Ground/sea reflection analysis (optional)
d) Plot Path Profile
e) Print Link Performance report

The output procedure in a different order of operation will operate on default values where they are necessary. The Ground/sea reflection analysis cannot operate without frequency and path profile data.

To obtain the rainfall rate, refer to the contour map of Australian and Fiji. Rainfall data for other regions can be used to calculate the 0.05% or 0.01% levels as required by the program.

2.1.1 LINK BUDGET BLOCK

The link budget block calculates the link signal levels and predicted attenuation and fading due to rain. One-way link availability is calculated for the worst month and annual period. All menu options except path geometry and Ground/sea reflection analysis need be selected to operate this block. Where appropriate, default values are set, pending execution of the other blocks.

At any time, the designer can edit any of the data that has been entered. This allows easy access to results using different equipments for example.

2.1.2 PATH GEOMETRY BLOCK

Input options can be selected on the actual data available. if insufficient data is available for output, RADLNK will advise the user.

Dot joining and earth bulge cannot be disabled under user control. Note, the degree of earth bulge is defined as the K Factor. The K Factor has a theoretical range of 0 thru infinity, however practical values in the program range 0.4 thru 10 (Default at 1.333).

A A3 size output is available with expanded X-Axis and Y-Axis. This provides higher resolution where clearances are marginal.

The CLR heading data, shows the clearance in metres between rayline and ground, the height of trees or buildings would need to be subtracted from the CLR data, if they were not considered in the path profile.

The rayline between stations must be drawn in by hand.

2.2 REFERENCE LIST

"Rain Attenuation Considerations for Satellite Paths in Australia" R.K. Flavin, Telecom Australia.

"Attenuation by Hydrometeors, Preciptation and other Atmospheric Particles" CCIT REP 721-1.

"Propagation data for design Terrestial, Troposcatter, Trans-Horizon and Earth-Space Telecommunication systems CCITT REC 530-1, REP 338-4.

"Antenna Theory. Analysis and Design" C A Balanis, Harper and Row.

"Digital Microwave Radio, Engineering Fundamentals" NEC, Tokyo. MSD-3003 820B-01.



FIGURE 1 – *Specific attenuation γ_R due to rain*

Raindrop size distribution [Laws and Parsons, 1943]
Terminal velocity of raindrops [Gunn and Kinzer, 1949]
Index of refraction of water at 20° C [Ray, 1972]
Spherical drops



FIG.7a WORST MONTH RAIN INTENSITY CONTOURS (mm/hr.)
FOR 0·1% PROBABILITY



FIG. 8a WORST MONTH RAIN INTENSITY CONTOURS (mm/hr.)
FOR 0·05% PROBABILITY



RAINFALL RATE EXCEEDED 0,01% OF THE YEAR.
(mm / hour)

NEW ZEALAND

PROGRAM ***RADLNK*** MICROWAVE LINK PREDICTIONS

RADIO LINK FROM TERMINAL [A] TO PASSIVE REPEATER

Path Length	5.700 Km
Earth Bulge @ K = 1.33	0.48 m
RMS Roughness @ K = 1.33	82.59 m
Max Fresnel Zone Radius	6.33 m

LINK SIGNAL LEVELS

Frequency	10.680 GHz
Free Space Path Loss	128.2 dB
Stn [A] Antenna Gain	39.4 dBi
Feedline Loss	0.4 dB
Stn [B] Antenna Gain	39.4 dBi
Feedline Loss	0.0 dB
Transmit Power	13.0 dBi
Atmospheric Attenuation	0.0 dB
MEDIAN RECEIVE LEVEL	-36.8 dB
Receive Threshold	-80.0 dBm
FADE MARGIN	43.2 dB

RADIO LINK FROM PASSIVE REPEATER TO TERMINAL [B]

Path Length	1.200 Km
Earth Bulge @ K = 1.33	0.02 m
RMS Roughness @ K = 1.33	80.21 m
Max Fresnel Zone Radius	2.91 m

LINK SIGNAL LEVELS

Frequency	10.680 GHz
Free Space Path Loss	114.7 dB
Stn [A] Antenna Gain	39.4 dBi
Feedline Loss	0.4 dB
Stn [B] Antenna Gain	39.4 dBi
Feedline Loss	0.4 dB
Transmit Power	-36.8 dBi
Atmospheric Attenuation	0.0 dB
MEDIAN RECEIVE LEVEL	-73.5 dB
Receive Threshold	-80.0 dBm
FADE MARGIN	6.5 dB

RAIN ATTENUATION OUTAGES

Climate Factor	1.2
Rain Rate Exceeds 0.050% /M	30.0 mm/H
Rain Attenuation	4.5 dB
RAIN MARGIN	2.1 dB
Rain Outage TIME/MONTH	0.0170 %
Rain Outage TIME/YEAR	0.0027 %

AMPLITUDE DISPERSION OUTAGES

Outage Time (Single Link)	0.0163 %

LINK AVAILABILITY (Excluding Hardware, One Way)

Availability (WORST MONTH)	99.9667 %
Availability (ANNUAL)	99.9810 %

PROGRAM ***RADLNK*** PRINT PROFILE DETAILS

PASSIVE REPEATER Support Height = 35 m
TERMINAL B Support Height = 35 m
Path Length (Km) = 1.2

TERRAIN PROFILE DATA

Data Point	1	DISTANCE From [A]	Km	0
		HEIGHT at this Point	m	760
Data Point	2	DISTANCE From [A]	Km	.5
		HEIGHT at this Point	m	660
Data Point	3	DISTANCE From [A]	Km	1.2
		HEIGHT at this Point	m	630

PROGRAM ***RADLNK*** PRINT PROFILE DETAILS

TERMINAL [A] Support Height = 35 m
PASSIVE REPEATER Support Height = 35 m
Path Length (Km) = 5.7

TERRAIN PROFILE DATA

Data Point	1	DISTANCE From [A]	Km	0
		HEIGHT at this Point	m	848
Data Point	2	DISTANCE From [A]	Km	.5
		HEIGHT at this Point	m	730
Data Point	3	DISTANCE From [A]	Km	1
		HEIGHT at this Point	m	700
Data Point	4	DISTANCE From [A]	Km	1.5
		HEIGHT at this Point	m	710
Data Point	5	DISTANCE From [A]	Km	2
		HEIGHT at this Point	m	650
Data Point	6	DISTANCE From [A]	Km	2.5
		HEIGHT at this Point	m	670
Data Point	7	DISTANCE From [A]	Km	3
		HEIGHT at this Point	m	660
Data Point	8	DISTANCE From [A]	Km	3.5
		HEIGHT at this Point	m	660
Data Point	9	DISTANCE From [A]	Km	4
		HEIGHT at this Point	m	690
Data Point	10	DISTANCE From [A]	Km	4.5
		HEIGHT at this Point	m	700
Data Point	11	DISTANCE From [A]	Km	5
		HEIGHT at this Point	m	630
Data Point	12	DISTANCE From [A]	Km	5.5
		HEIGHT at this Point	m	730
Data Point	13	DISTANCE From [A]	Km	5.7
		HEIGHT at this Point	m	760

PROGRAM ***PRAD*** Engsoft MICROWAVE LINK PREDICTIONS

RADIO LINK FROM PENSHURST TO HAMILTON

Path Length	24.200 Km
Earth Bulge [K = 1.33	8.62 m
RMS Roughness [K = 1.33	70.60 m
Max Fresnel Zone Radius	11.72 m

LINK SIGNAL LEVELS

Frequency	13.250 GHz
Free Space Path Loss	142.5 dB
Stn `A§ Antenna Gain	34.0 dBi
Feedline Loss	0.0 dB
Stn `B§ Antenna Gain	34.0 dBi
Feedline Loss	0.0 dB
Transmit Power	22.0 dBi
Atmospheric Attenuation	0.2 dB
MEDIAN RECEIVE LEVEL	-52.7 dB
Receive Threshold	-80.0 dBm
FADE MARGIN	27.3 dB

RAIN ATTENUATION OUTAGES

Climate Factor	1.0
Rain Rate Exceeds 0.050% /M	15.0 mm/H
Rain Attenuation	11.5 dB
RAIN MARGIN	15.8 dB
Rain Outage TIME/MONTH	0.0046 %
Rain Outage TIME/YEAR	0.0006 %

AMPLITUDE DISPERSION OUTAGES

Outage Time (Single Link)	0.0063 %

LINK AVAILABILITY (Excluding Hardware, One Way)

Availability (WORST MONTH)	99.9891 %
Availability (ANNUAL)	99.9931 %

PROGRAM ***RADLNK*** MICROWAVE LINK PREDICTIONS

SYSTEM DETAILS

FREQUENCY	10.680 GHZ
PATH LENGTH	20.700 KM
STATION A	DARKES FOREST
STATION B	ENGADINE
EQUIPMENT TYPE	MINILINK 10

TERRAIN DATA — PERFORMED AT K = 1.3333

EARTH BULGE	6.3 M
RMS ROUGHNESS	42.0 M
FRESNEL ZONE RADIUS	12.1 M

LINK SIGNAL LEVELS

FREE SPACE PATH LOSS	139.4 DB
TRANSMIT ANTENNA GAIN	32.0 DBI
RECEIVE ANTENNA GAIN	32.0 DBI
TRANSMIT POWER	13.0 DBM
WAVEGUIDE/OTHER LOSSES	0.0 DB
RECEIVE LEVEL	-62.4 DBM
RECEIVE THRESHOLD	-80.0 DBM
FADE MARGIN	17.6 DB

RAIN ATTENUATION OUTAGES

CLIMATE FACTOR	1.0
RAIN RATE EXCEEDED 0.05%	30.0 MM/H
RAIN ATTENUATION	13.5 DB
RAIN MARGIN	4.1 DB
RAIN OUTAGE TIME/MONTH	0.0947 %
RAIN OUTAGE TIME/YEAR	0.0216 %

AMPLITUDE DISPERSION OUTAGES

OUTAGE TIME (SINGLE LINK)	0.0300 %
SPACE DIVERSITY SEPARATION	8.0 M
OUTAGE TIME (DIVERSITY LINK)	0.0120 %

LINK AVAILABILITY (EXCLUDING HARDWARE)

AVAILABILITY (WORST MONTH)	99.8932 %
AVAILABILITY (ANNUAL)	99.9664 %

PROGRAM ***RADLNK*** RADIO PATH DETAILS

DARKES FOREST SUPPORT HEIGHT = 20 M
ENGADINE SUPPORT HEIGHT = 15 M
PATH LENGTH (KM) = 20.7

TERRAIN PROFILE DATA
DATA POINT 1
DISTANCE FROM A (KM) 0
HEIGHT AT THIS POINT (M) 380
DATA POINT 2
DISTANCE FROM A (KM) 3
HEIGHT AT THIS POINT (M) 300
DATA POINT 3
DISTANCE FROM A (KM) 4
HEIGHT AT THIS POINT (M) 300
DATA POINT 4
DISTANCE FROM A (KM) 5
HEIGHT AT THIS POINT (M) 280
DATA POINT 5
DISTANCE FROM A (KM) 6.3
HEIGHT AT THIS POINT (M) 310
DATA POINT 6
DISTANCE FROM A (KM) 7.5
HEIGHT AT THIS POINT (M) 280
DATA POINT 7
DISTANCE FROM A (KM) 8
HEIGHT AT THIS POINT (M) 260
DATA POINT 8
DISTANCE FROM A (KM) 8.8
HEIGHT AT THIS POINT (M) 200
DATA POINT 9
DISTANCE FROM A (KM) 10.5
HEIGHT AT THIS POINT (M) 240
DATA POINT 10
DISTANCE FROM A (KM) 12.6
HEIGHT AT THIS POINT (M) 200
DATA POINT 11
DISTANCE FROM A (KM) 13.5
HEIGHT AT THIS POINT (M) 120
DATA POINT 12
DISTANCE FROM A (KM) 14.3
HEIGHT AT THIS POINT (M) 200
DATA POINT 13
DISTANCE FROM A (KM) 19.5
HEIGHT AT THIS POINT (M) 100
DATA POINT 14
DISTANCE FROM A (KM) 20.7
HEIGHT AT THIS POINT (M) 200

RAIN RATE 0.05% PER MONTH = 30
CLIMATE ZONE = 1

PROGRAM *RADLNK* GROUND/SEA REFLECTION ANALYSIS

REGION FOR CONSIDERATION IS ILLUMINATED PATH
LINK FROM DARKES FOREST TO ENGADINE

K FACTOR 0.600
REFLECTION POINT AT 1.294 KM FROM A. RAY CLEARANCE 36.174 M
SD SEPARATION. STN A 0.8 M
STN B 12.5 M

REFLECTION POINT AT 1.553 KM FROM A. RAY CLEARANCE 39.924 M
SD SEPARATION. STN A 0.9 M
STN B 10.7 M

SCAN FOR CLEARANCE < 2 FRESNEL ZONES
SCAN COMPLETE

K FACTOR 1.333
REFLECTION POINT AT 1.294 KM FROM A. RAY CLEARANCE 37.980 M
SD SEPARATION. STN A 0.8 M
STN B 12.4 M

SCAN FOR CLEARANCE < 2 FRESNEL ZONES
SCAN COMPLETE

K FACTOR 10.000
REFLECTION POINT AT 1.035 KM FROM A. RAY CLEARANCE 34.944 M
SD SEPARATION. STN A 0.1 M
STN B 1.3 M

SCAN FOR CLEARANCE < 2 FRESNEL ZONES
SCAN COMPLETE

TERRAIN PROFILE DARKES FOREST TO ENGADINE



K FACTOR = 1.33 RMS ROUGHNESS = 42.00

TERRAIN PROFILE DARKES FOREST TO ENGADINE



```
100 REM     PROGRAM          ****RADLNK****          MICROWAVE RADIO PREDICTIONS
105 REM  A. SCOPE
110 REM     THE PROGRAM CALCULATES  1 - PATH LOSS  (UNOBSTRUCTED)
115 REM                             2 - RAIN ATTENUATION       2GHZ<F<99GHZ
120 REM                             3 - RAIN OUTAGE TIMES
125 REM                             4 - AMPLITUDE DISPERSION FADING
130 REM                             5 - FADE MARGIN GIVEN EQUIPMENT SPEC'S
135 REM                             6 - BEARINGS & DISTANCES OF STATIONS
140 REM                             7 - FREQUENCY & SPACE DIVERSITY DATA
145 REM                             8 - FRESNEL ZONE SIZE & EARTH BULGE
150 REM                             9 - RADIO HORIZON    FOR INTERFERENCE
155 REM                            10 - REFLECTION POINTS ALONG THE ROUTE
160 REM
165 REM     THE PROGRAM OUTPUTS     1 - SYSTEM DETAILS
170 REM                             2 - TERRAIN DATA       (FOR PATH PLOT)
175 REM                             3 - LINK SIGNAL LEVELS
180 REM                             4 - RAIN OUTAGE TIMES
185 REM                             5 - NORMAL FADING OUTAGE TIMES
190 REM                             6 - SYSTEM AVAILABILITY,   NO HARDWARE
195 REM                             7 - PATH PLOT (A3 & A4 PLOT SIZES)
200 REM                             8 - PATH REFLECTIVITY AT 3 K VALUES
205 REM                             9 - RECOMMENDED SPACE DIVERSITY SEP
210 REM                            10 - EQUIPMENT DATABASE HANDLER.
215 REM  B. OPERATION
220 REM     FOR THE INITIAL RUN, IT IS NECESSARY TO LOAD THE PROGRAM WITH
225 REM     DATA. THIS IS PERFORMED BY SELECTING ALL MAIN MENU OPTIONS
230 REM     (IN ANY SEQUENCE) PRIOR TO PRINTING THE REPORT (5).
235 REM     THE OPTION 'PATH GEOMETRY' AND IT'S SUB-MENU ITEMS DOES NOT
240 REM     HAVE TO BE USED INITIALLY AS DEFAULT VALUES ARE SET.
245 REM     SUBSEQUENT CHANGES CAN BE MADE BY SELECTING ANY OF THE MAIN
250 REM     MENU OPTIONS AND THEN PRINTING THE NEW REPORT (OR PLOT).
255 REM     THE PATH PROFILE PLOT OPTION IS INDEPENDANT OF THE REST OF THE
260 REM     PROGRAM. THREE VARIABLES ARE INTERLINKED, REFER TO 2.1 PROGRAM
265 REM     OPERATION. THIS ALLOWS THE USER TO EVALUATE THE PATH OBSTACLES.
270 REM     MINIMIZATION OF RADIO OVERSHOOT WILL ALSO REDUCE POSSIBLE
275 REM     INTERFERENCE.
280 REM
285 REM  C. PROGRAM CAN BE RESET BY EXITING AND RESTARTING. PATH GEOMETRY
290 REM     DATA CAN BE RESET WITHIN IT'S OWN SELECTION.     ALTERNATIVELY
295 REM     PROGRAM DATA MAY BE OVER-WRITTEN.
300 REM
305 REM  D. THE PROGRAM ASSUMES UNOBSTRUCTED PATHS FOR THE RADIO RAYLINE.
310 REM     REFER TO FRESNEL ZONE CLEARANCE RULES IN 1.1.1 DESIGNING LINKS
315 REM
320 REM  E. RAIN RATES ARE FOR 0.05% PROBABILITY FOR THE WORST MONTH.
325 REM     REF:  CCITT REP 563  AND HANDBOOK RAIN CONTOURS CHARTS.
330 REM
335 REM  F. CONVERSION OF UNITS:   DBM=10*LOG(WATTS/0.001)
340 REM                            DBI=DBD + 2.12
345 REM
350 REM  G. PROGRAM WRITTEN BY: DAVID BURGER     JULY 1985
355 REM
360 DIM X(101),Y(101),XD(101),YD(101),RF(101)
365 T46=27:KF=1.3333:R2=13:R3=1:R44=.05:R4$="M"      'DEFAULT VALUE SET
```

```
500 CLS:PRINT"PROGRAM        ***RADLNK***                    MICROWAVE LINK PREDICTI
ONS"
505 SOUND 88,4:SOUND 151,3:SOUND 88,2
510 PRINT TAB(44)"D.BURGER VS 2.4":PRINT:PRINT:PRINT:PRINT
515 PRINT TAB(22)"1 - INPUT  SYSTEM   DETAILS"
520 PRINT TAB(22)"2 - PATH  PROFILE  GEOMETRY"
525 PRINT TAB(22)"3 - INPUT  RAIN INTENSITIES"
530 PRINT TAB(22)"4 - EQUIPMENT    PARAMETERS"
535 PRINT TAB(22)"5 - LINK PERFORMANCE REPORT":PRINT TAB(22)"6 - DEFAULT PARAMET
ERS":PRINT TAB(22)"7 - EXIT"
540 KEY OFF
545 QX$=INKEY$:IF QX$="" THEN 545
550 IF VAL(QX$)>7 OR VAL(QX$)<1 THEN 545
555 SOUND 88,3
560 ON VAL(QX$) GOSUB 700,3800,1000,1200,2000,4700,7500
565 GOTO 500
570 END
```

```
700 REM      ************************SYSTEM DETAILS
705 CLS:PRINT"PROGRAM      ***RADLNK***          SYSTEM DETAILS INPUT"
710 PRINT:PRINT:PRINT"TO RETURN TO MAIN MENU  <EXIT>":PRINT
715 PRINT"<CURRENT DATA>":PRINT"STATION  A ";N1$
720 PRINT"STATION  B ";N2$
725 PRINT
730 INPUT"TERMINAL  A  NAME ";QX$
735 IF QX$="" THEN 745
740 IF QX$="EXIT" THEN RETURN ELSE N1$=QX$
745 INPUT"TERMINAL  B  NAME ";QX$
750 IF QX$="" THEN 760
755 IF QX$="EXIT" THEN RETURN ELSE N2$=QX$
760 FLAG2=0:PRINT:PRINT"ENTER GRID COORDINATES   Y/N"
765 QX$=INKEY$:IF QX$="" THEN 765
770 IF QX$="Y" THEN 3500
775 PRINT"DISTANCE BETWEEN TERMINALS   KM ";P1;"    ";:INPUT QX$
780 IF QX$="" THEN 795
785 IF QX$="EXIT" THEN RETURN ELSE P1=VAL(QX$)
790 IF P1<=0 THEN PRINT"<ERROR>":SOUND 88,3:GOTO 775
795 PRINT:PRINT"1 - NO DIVERSITY":PRINT"2 - SPACE DIVERSITY"
800 PRINT"3 - FREQUENCY DIVERSITY":PRINT:PRINT"INPUT DIVERSITY OPTION"
805 QX$=INKEY$:IF QX$="" THEN 805
810 IF QX$="EXIT" THEN RETURN ELSE FLAG3=VAL(QX$)
815 IF FLAG3<2 OR FLAG3>3 THEN 860
820 IF FLAG3=2 THEN 825 ELSE 845
825 PRINT"VERTICAL  SEPARATION    (M) ";SD1;:INPUT "  ";QX$
830 IF QX$="" THEN 835 ELSE SD1=ABS(VAL(QX$))
835 IF SD1<.01 THEN PRINT"<ERROR>":SOUND 88,3:GOTO 825
840 GOTO 860
845 PRINT"FREQUENCY SEPARATION  (MHZ) ";FD1;:INPUT "  ";QX$
850 IF QX$="" THEN 855 ELSE FD1=ABS(VAL(QX$))
855 IF FD1<7 THEN PRINT"<ERROR>":SOUND 88,3:GOTO 845
860 RETURN
```

```
1200 REM      ************************EQUIPMENT PARAMETERS CONTROL
1205 CLS:PRINT"PROGRAM      ***RADLNK***              EQUIPMENT PARAMETERS INPUT"
1210 PRINT:Z$="NOT IN DATABASE"
1215 PRINT:PRINT:PRINT
1220 PRINT TAB(17)"KEYBOARD INPUT":PRINT TAB(22)"1 - SPECIFY  DATA":PRINT
1225 PRINT TAB(17)"FILE INPUT":PRINT TAB(22)"2 - VIEW DATA BASE"
1230 PRINT TAB(22)"3 - MODIFY EQUIPMENT DATABASE":PRINT TAB(22)"4 - WRITE NEW EQ
PT. DATABASE"
1235 PRINT TAB(22)"5 - PRINT EQUIPMENT LIST":PRINT:PRINT TAB(17)"EXIT":PRINT TAB
(22)"6 - EXIT TO MAIN MENU"
1240 QX$=INKEY$:IF QX$="" THEN 1240
1245 IF VAL(QX$)<1 OR VAL(QX$)>6 THEN 1240
1250 IF VAL(QX$)=6 THEN RETURN
1251 IF VAL(QX$)=1 THEN 1265
1252 PRINT"CURRENT EQUIPMENT FILES"
1253 FILES "*.EQP"
1255 PRINT:INPUT"INPUT EQUIPMENT MANUFACTURER ";Z$
1260 Z$=LEFT$(Z$,5)+".EQP"
1265 ON VAL(QX$) GOSUB 1500,2500,3000,2600,2800
1270 GOTO 1205
```

```
1500 REM      ************************SPECIFY EQUIPMENT PARAMETERS
1505 CLS:PRINT"PROGRAM      ***RADLNK***          SPECIFY EQUIPMENT PARAMETERS"
1510 PRINT"EQUIPMENT MANUFACTURER FILE: ";Z$
1515 PRINT:PRINT" ";N3$:PRINT" FREQUENCY    ";F1:PRINT" TX POWER     ";P2
1520 PRINT" RX THRESHOLD ";R1:PRINT" TX ANT GAIN  ";G1:PRINT" RX ANT GAIN  ";G2
1525 PRINT" WAVEGUIDE LOSS ";G3
1530 INPUT"EQUIPMENT  TYPE        ";QX$
1535 IF QX$="" THEN 1545
1540 IF QX$="EXIT" THEN RETURN ELSE N3$=QX$
1545 INPUT"SIGNAL  FREQUENCY  GHZ ";QX$
1550 IF QX$<>"" THEN F1=ABS(VAL(QX$))
1555 INPUT"TRANSMIT  POWER    DBM ";QX$
1560 IF QX$<>"" THEN P2=VAL(QX$)
1565 INPUT"RECEIVE THRESHOLD  DBM ";QX$
1570 IF QX$<>"" THEN R1=VAL(QX$)
1575 INPUT"TOTAL WAVEGUIDE LOSSES ";G3
1580 PRINT:PRINT"<INPUT DISH SIZE  (M)   Y/N"
1585 QX$=INKEY$:IF QX$="" THEN 1585
1590 IF QX$="Y" THEN 1625
1595 INPUT"TRANSMIT ANTENNA  GAIN (DBI)";QX$
1600 IF QX$<>"" THEN G1=VAL(QX$)
1605 INPUT"RECEIVE  ANTENNA  GAIN (DBI)";QX$
1610 IF QX$<>"" THEN G2=VAL(QX$)
1615 N5$=" "
1620 RETURN
1625 INPUT"SIZE OF BOTH DISH ANTENNAS  (M) :";DD
1630 G1=10*LOG(4.78*(ABS(DD)*F1/.3)^2)/LOG(10)       'DISH ANTENNA GAIN
1635 N5$=" +"+STR$(DD)+"M DISH"
1640 G2=G1
1645 RETURN
```

```
2000 REM      ************************LINK PERFORMANCE REPORT
2005 IF P1<=0 OR F1<=0 THEN PRINT"<NO DATA>":SOUND 88,3:GOSUB 4100:RETURN
2010 P9=92.5+20*LOG(P1)/LOG(10)+20*LOG(F1)/LOG(10)
2013 GOSUB 3300                                          'ATMOSPHERIC ATTENUATION
2015 R9=P2+G1+G2-P9-G3-ATMOS
2020 F9=R9-R1
2025 GOSUB 3200                                          'AMPLITUDE DISP FADING
2030 T44=P1*P1/(KF*50.96)
2035 T48=17.34*SQR(P1/(4*F1))
2040 O2=O1
2045 IF FLAG3=3 THEN 2050 ELSE 2060
2050 IF O1>=.05 THEN X=SQR(O1)/(.18*FD1/F1) ELSE X=O1/(.035*FD1/F1)
2055 GOTO 2065
2060 IF FLAG3=2 THEN X=830*P1*10^(-F9/10)/(F1*SD1*SD1) ELSE 2070
2065 O2=O1*X/SQR(1+X*X)
2070 CLS
2075 GOSUB 3240                                          'RAIN FADING
2080 O8=100-O2-PRY
2085 O9=100-O2-PRM
2090 F8=F9-A8
2095 IF O1>100 THEN O1=100
2100 IF O2>100 THEN O2=100
2105 IF O9<0 THEN O9=0
2110 IF O8<0 THEN O8=0
2115 PRINT"SUBROUTINE    *LINK PERFORMANCE PRINTOUT*"
2120 LPRINT"PROGRAM       ***RADLNK***          MICROWAVE LINK PREDICTIONS"
2125 LPRINT:LPRINT"SYSTEM DETAILS"
2130 LPRINT USING "     Frequency                    ###.###";F1;:LPRINT" GHz"
2135 LPRINT USING "     Path Length                  ###.###";P1;:LPRINT" Km"
2140 IF FLAG2=1 THEN LPRINT USING"     Bearing of STATION B         ###.#";H;:LP
RINT"  DEG"
2145 LPRINT"     STATION  A                    ";N1$
2150 IF FLAG2=1 THEN LPRINT"              Latitude & Longitude ";
2155 IF FLAG2=1 THEN LPRINT USING"  ####.##  ";AA1,AA2
2160 LPRINT"     STATION  B                    ";N2$
2165 IF FLAG2=1 THEN LPRINT"              Latitude & Longitude ";
2170 IF FLAG2=1 THEN LPRINT USING"  ####.##  ";BB1,BB2
2175 LPRINT"     Equipment Type                ";N3$+N5$
2180 LPRINT:LPRINT"TERRAIN DATA";TAB(36)"PERFORMED AT K = ";KF
2185 LPRINT USING "     Earth Bulge                  ###.#";T44;:LPRINT" m"
2190 LPRINT USING "     RMS Roughness                ###.#";T46;:LPRINT" m"
2195 LPRINT USING "     Fresnel Zone Radius          ###.#";T48;:LPRINT" m"
2200 LPRINT:LPRINT"LINK SIGNAL LEVELS"
2205 LPRINT USING "     Free Space Path Loss        ####.#";P9;:LPRINT" dB"
2210 LPRINT USING "     Transmit Antenna Gain        ###.#";G1;:LPRINT" dBi"
2215 LPRINT USING "     Receive  Antenna Gain        ###.#";G2;:LPRINT" dBi"
2220 LPRINT USING "     Transmit Power               ###.#";P2;:LPRINT" dBm"
2225 LPRINT USING "     Waveguide + Other Losses     ###.#";G3;:LPRINT" dB"
2227 LPRINT USING "     Atmospheric Attenuation     ####.#";ATMOS;:LPRINT" dB"
2230 LPRINT:LPRINT USING "     MEDIAN RECEIVE LEVEL         ####.#";R9;
2235 LPRINT" dBm"
2240 LPRINT USING "     Receive Threshold            ###.#";R1;:LPRINT" dBm"
2245 LPRINT:LPRINT USING "     FADE MARGIN                  ####.#";F9;
2250 LPRINT" dB"
2255 LPRINT:LPRINT:LPRINT"RAIN ATTENUATION OUTAGES"
2260 LPRINT USING "     Climate Factor                 #.#";R3
```

```
2800 REM      **************************PRINT EQUIPMENT LIST
2805 LPRINT"PROGRAM           ***RADLNK***             EQUIPMENT LIST"
2810 LPRINT:LPRINT"Manufacturer ";Z$
2815 LPRINT"    TYPE";TAB(28);"F GHz";TAB(35);"Tx POW";TAB(45);"Rx THRES";TAB(55
);"ANT Gain";TAB(65);"RF Loss's"
2820 OPEN Z$ FOR INPUT AS #1
2825 INPUT #1,N3$,G3,P2,R1,G1,F1
2830 IF LEFT$(N3$,3)="END" THEN 2855
2835 LPRINT TAB(5);N3$;TAB(26);:LPRINT USING"###.###";F1;
2840 LPRINT USING"   ###.#";P2;:LPRINT USING"      ####.#";R1;
2845 LPRINT USING"     ###.#";G1;:LPRINT USING"      ###.#";G3
2850 GOTO 2825
2855 CLOSE #1
2860 LPRINT:LPRINT
2865 RETURN
```

```
3000 REM      **********************MODIFY EQUIPMENT FILE
3005 CLS:FLAG5=0
3010 OPEN "DUMP.EQP" FOR OUTPUT AS #2
3015 OPEN Z$ FOR INPUT AS #1
3020 INPUT #1,N3$,G3,P2,R1,G1,F1
3025 IF LEFT$(N3$,3)="END" THEN FLAG5=1
3030 GOSUB 1500                                        'SPECIFY EQUIPMENT
3035 WRITE #2,N3$,G3,P2,R1,G1,F1
3040 PRINT:PRINT"<MORE DATA ?    Y/N >"
3045 QX$=INKEY$:IF QX$="" THEN 3045
3050 IF QX$="N" THEN 3055 ELSE IF FLAG5=1 THEN 3030 ELSE 3020
3055 IF FLAG5=0 THEN 3060 ELSE 3075
3060 INPUT #1,N3$,G3,P2,R1,G1,F1
3065 WRITE #2,N3$,G3,P2,R1,G1,F1
3070 IF LEFT$(N3$,3)="END" THEN 3080 ELSE 3060
3075 WRITE #2,"END",G3,P2,R1,G1,F1
3080 CLOSE #2
3085 CLOSE #1
3090 REM  SWAP FILES
3095 OPEN Z$ FOR OUTPUT AS #1
3100 OPEN "DUMP.EQP" FOR INPUT AS #2
3105 INPUT #2,N3$,G3,P2,R1,G1,F1
3110 WRITE #1,N3$,G3,P2,R1,G1,F1
3115 IF LEFT$(N3$,3)="END" THEN 3120 ELSE 3105
3120 CLOSE #1:CLOSE #2
3125 KILL"DUMP.EQP"
3130 RETURN
```

```
3200 REM    ***********AMPLITUDE DISPERSION FADING
3205 IF T46<6 THEN T4=6
3210 IF T46>42 THEN T4=42 ELSE T4=T46
3215 O1=.002088*(10^(-F9/20))^2*F1*(P1^3)*(T4^-1.27)*R3
3220 IF O1>=99 THEN 3230
3225 RETURN
3230 O1=100-2.37*(T4^-1.27)*R3*10^(F9/7.88)/LOG(P1/10)
3235 RETURN



3240 REM    ***********RAIN FADING
3245 IF F1>75 THEN DU=75+.3*(F1-75) ELSE DU=F1
3250 A8=.495*LOG(R2)-1.931+(2.847-.1911*LOG(R2))*LOG(.4344*LOG(DU))
3255 A8=P1*10^A8
3260 IF R4$="M" THEN R4=R44^1.15*.29 ELSE R4=R44
3265 IF A8<F9 THEN PRY=R4*(ABS(A8/F9))^3.2 ELSE PRY=R4*(ABS(A8/F9))^2.6
3270 PRM=(3.45*PRY)^.87
3275 RETURN



3300 REM    ***********ATMOSPHERIC ATTENUATION
3305 IF F1>57 THEN 3320
3310 A8=(6.6/(F1^2+.33)+9/((ABS(F1-57))^2+1.96))*F1^2*.001
3315 GOTO 3335
3320 IF F1<63 THEN A8=14.9:GOTO 3335 ELSE 3325
3325 IF F1>350 THEN A8=100:GOTO 3335
3330 A8=(4.13/(ABS(F1-63)^2+1.1)+.19/(ABS(F1-118.7)^2+2))*F1^2*.001
3335 ATMOS=P1*A8
3340 RETURN
```

```
3500 REM   ***********BEARINGS & DISTANCE
3505 REM OUTPUT     H=HEADING(DEGREES)      P1=DISTANCE(KILOMETERS)
3510 PRINT:PRINT"            *BEARINGS & DISTANCE*"
3515 PRINT"      SOUTH LATITUDES ENTER AS -VE"
3520 PRINT"      EAST LONGITUDES ENTER AS -VE"
3525 PI#=3.1415927#:FLAG2=1
3530 INPUT"      TERMINAL A     LAT,LONG :";AA1,AA2
3535 A1#=AA1/360*2*PI#
3540 A2#=AA2/360*2*PI#
3545 INPUT"      TERMINAL B     LAT,LONG :";BB1,BB2
3550 B1#=BB1/360*2*PI#
3555 B2#=BB2/360*2*PI#
3560 A3#=B2#-A2#
3565 A4#=SIN(A1#)*SIN(B1#)+COS(A1#)*COS(B1#)*COS(A3#)
3570 A5#=SQR(1-A4#*A4#)
3575 A6#=A5#/A4#:A7#=ATN(A6#)
3580 IF A6#<0# THEN 3585 ELSE 3590
3585 A7#=PI#+A7#
3590 D#=60*360*A7#/(2*PI#)
3595 P1=D#*1.853:S=D#*1.152
3600 H1#=(SIN(B1#)-SIN(A1#)*COS(A7#))/(SIN(A7#)*COS(A1#))
3605 IF H1#*H1#=1# THEN H2#=0# ELSE H2#=SQR(1#-H1#*H1#)
3610 H3#=H2#/H1#
3615 H4=ATN(H3#)
3620 IF H3<0 THEN 3625 ELSE 3630
3625 H4=PI+H4
3630 H5=H4*57.2957795#
3635 IF SIN(A3)<0 THEN H=H5 ELSE H=360-H5
3640 GOTO 795
```

```
3800 REM    ****B*U*R*G*E*R*****PATH GEOMETRY
3805 CLS:PRINT"PROGRAM       ***RADLNK***          PATH GEOMETRY & EVALUATION"
3810 PRINT:PRINT:PRINT:PRINT:PRINT TAB(22)"1 - ENTER PATH PROFILE"
3815 PRINT TAB(22)"2 - ENTER SUPPORT HEIGHTS"
3820 PRINT TAB(22)"3 - PLOT A3 SIZE":PRINT TAB(22)"4 - PLOT A4 SIZE"
3825 PRINT TAB(22)"5 - PRINT PATH INFORMATION":PRINT TAB(22)"6 - GROUND/SEA REFL
ECTION ANALYSIS"
3830 PRINT TAB(22)"7 - EXIT TO MAIN MENU":PRINT:PRINT
3835 QX$=INKEY$:IF QX$="" THEN 3835
3840 IF VAL(QX$)<1 OR VAL(QX$)>7 THEN 3835
3845 SOUND 100,1:IF VAL(QX$)=7 THEN RETURN
3850 IF VAL(QX$)=3 THEN U5=80:U6=80 ELSE U5=35:U6=50
3860 ON VAL(QX$) GOSUB 4200,4000,5200,5200,5000,6500
3865 GOTO 3805
```

```
4000 REM  ********************SUPPORT HEIGHTS
4005 CLS:PRINT"PROGRAM       ***RADLNK***            STATION SUPPORT HEIGHT [AGL]"
4010 PRINT:PRINT"EXISTING SUPPORT HEIGHT AT  ";N1$;TAB(44);" =";D1;" M    [A]"
4015 PRINT"EXISTING SUPPORT HEIGHT AT  ";N2$;TAB(44);" =";D2;" M    [B]"
4020 PRINT:INPUT"HEIGHT OF STATION  [A]  SUPPORT       (M)";QX$
4025 IF QX$="" THEN 4030 ELSE D1=ABS(VAL(QX$))
4030 INPUT"HEIGHT OF STATION  [B]  SUPPORT       (M)";QX$
4035 IF QX$="" THEN 4040 ELSE D2=ABS(VAL(QX$))
4040 RH2=4.12*(SQR(D1)+SQR(D2))
4045 PRINT"<RADIO HORIZON  (SMOOTH EARTH) =";RH2;" KM"
4050 GOSUB 4100                                          'WAIT
4055 RETURN


4100 REM       ************************WAIT
4105 FOR COUNT=1 TO 655:NEXT COUNT
4110 RETURN
```

```
4200 REM ******************PATH PROFILE ENTRY
4205 CLS:PRINT"PROGRAM       ***RADLNK***           CONTOUR PROFILE ENTRY  [ASL]"
4210 PRINT:PRINT:PRINT"THE HEIGHT OF EACH STATION (ABOVE DATUM) IS ENTERED"
4215 PRINT"AT ZERO DISTANCE FROM STATION [A]. STATION [B] DATUM HEIGHT"
4220 PRINT"IS ENTERED AT THE ACTUAL PATH LENGTH DISTANCE":PRINT
4225 IF P1<=.0003 THEN INPUT"DISTANCE BETWEEN TERMINALS   (KM) ";P1 ELSE PRINT"P
ATH DISTANCE = ";P1;" KM"
4230 PRINT:PRINT"<EDIT PATH DATA >   Y/N"
4235 QX$=INKEY$:IF QX$="" THEN 4235
4240 IF QX$="Y" THEN GOSUB 4500
4245 PRINT:PRINT"TYPE 'EXIT' WHEN ALL DATA POINTS ARE ENTERED":PRINT
4250 PRINT"DATA POINT ";N+1:INPUT"     DISTANCE  FROM  [A]   KM ";QX$
4255 IF QX$="" THEN 4250
4260 IF QX$="EXIT" THEN 4290
4265 IF VAL(QX$)<0 OR VAL(QX$)>(P1+.001) THEN PRINT"<ERROR>":SOUND 88,3:GOTO 425
0 ELSE N=N+1
4270 XD(N)=VAL(QX$)
4275 INPUT"     HEIGHT AT THIS POINT   M ";YD(N)
4280 PRINT:PRINT
4285 GOTO 4250
4290 FOR J=1 TO N                                   'ORDER INPUT DATA
4295 FOR I=2 TO N
4300 IF XD(I)>XD(I-1) THEN 4320
4305 DUMPX=XD(I-1):DUMPY=YD(I-1)
4310 XD(I-1)=XD(I):YD(I-1)=YD(I)
4315 XD(I)=DUMPX:YD(I)=DUMPY
4320 NEXT I
4325 NEXT J
4330 RETURN
```

```
4500 REM  ****************PATH DATA EDITOR
4505 CLS:PRINT"PROGRAM       ***RADLNK***          PATH PROFILE DATA EDITOR"
4510 PRINT:PRINT"POINT","DIST","HEIGHT"
4515 FOR I=1 TO N:PRINT I,XD(I),YD(I):NEXT I
4520 PRINT:PRINT"(E)DIT  (R)ESET   E(X)IT"
4525 QX$=INKEY$:IF QX$="" THEN 4525
4530 IF QX$="R" THEN 4560
4535 IF QX$="E" THEN INPUT"INPUT POINT,DIST,HEIGHT";J,XJ,YJ ELSE 4290
4540 IF J>N OR J<0 THEN PRINT"<ERROR>":SOUND 88,4:GOTO 4505
4545 IF XD(J)>P1+.0001 THEN PRINT"<ERROR>":SOUND 88,4:GOTO 4505
4550 IF XD(J)<0 THEN PRINT"<ERROR>":SOUND 88,4:GOTO 4505 ELSE XD(J)=XJ:YD(J)=YJ
4555 GOTO 4505
4560 FOR I=1 TO N
4565 XD(I)=0:YD(I)=0
4570 NEXT I
4575 FOR I=1 TO 101
4580 X(I)=0:Y(I)=0:RF(I)=0
4585 NEXT I
4590 N=0
4595 RETURN
```

```
4700 REM  ****************DEFAULT  VALUES
4705 CLS:PRINT"PROGRAM       ***RADLNK***          DEFAULT PARAMETERS SET"
4710 PRINT:PRINT:PRINT"RMS ROUGHNESS (M) ";TAB(40);T46;TAB(55);
4715 INPUT QX$
4720 IF QX$="" THEN 4725 ELSE T46=ABS(VAL(QX$))
4725 PRINT:PRINT"K FACTOR           ";TAB(40);KF;TAB(55);
4730 INPUT QX$
4735 IF QX$="" THEN 4740 ELSE KF=VAL(QX$)
4740 IF KF<.4 THEN KF=.6
4745 IF KF>10 THEN KF=10
4750 PRINT:PRINT"RAINFALL RATE PERCENTILE DATA ";TAB(40);R44;TAB(55);
4755 INPUT QX$
4760 IF QX$="" THEN 4765 ELSE R44=ABS(VAL(QX$))
4765 PRINT:PRINT"RAINFALL BASED MONTH OR YEAR  M/Y ";TAB(42);R4$;TAB(55);
4770 INPUT QX$
4775 IF QX$="" THEN 4785
4780 IF QX$="Y" OR QX$="M" THEN R4$=QX$:GOTO 4785 ELSE PRINT"<ERROR>":SOUND 88,3
:GOTO 4765
4785 RETURN
```

```
5500 REM    ********************DOT JOINING - DATA FILL
5505 PRINT                'NYQUIST SAMPLING OF INPUT DATA
5510 FOR I=1 TO 101:X(I)=0:Y(I)=0:RF(I)=0:NEXT I:PRINT"<WAIT>"
5515 FOR I=1 TO U5+1
5520 KN=0
5525 FOR J=1 TO N
5530 X(I)=R*(I-1)
5535 IF ABS(XD(J)-X(I))<.51*R AND KN=0 THEN Y(I)=YD(J):KN=1:GOTO 5545
5540 IF ABS(XD(J)-X(I))<.51*R AND KN=1 THEN Y(I+1)=YD(J)
5545 NEXT J
5550 NEXT I
5555 I=0
5560 REM    DOT JOINING
5565 I=I+1
5570 IF I>U5+1 THEN PRINT"<DOT    JOINING      COMPLETE>":RETURN
5575 IF Y(I)=0 THEN 5565
5580 J=I
5585 J=J+1
5590 IF J>U5+1 THEN 5600
5595 IF Y(J)=0 THEN 5585
5600 FOR K=I TO J-2
5605 Y(K+1)=Y(K)-(Y(I)-Y(J))/(J-I)
5610 NEXT K
5615 I=J-1
5620 GOTO 5565
```

```
5700 REM    **********EARTH BULGE
5705 REM    PERFORMED AT  K
5710 FOR I=2 TO U5
5715 DS=P1/U5*(I-1)
5720 Y(I)=Y(I)+(DS*P1-DS*DS)/(KF*12.74)
5725 NEXT I
5730 PRINT"<EARTH BULGE ADJUSTMENT COMPLETE>"
5735 RETURN
```

```
5800 REM    **********RMS ROUGHNESS
5805 M=0:B=0:C=0:T=0
5810 FOR I=2 TO U5
5815 RF(I)=D1+(D1-D2)/(I/U5)-Y(I)
5820 NEXT I
5825 FOR I=2 TO U5
5830 C=C+RF(I):B=B+RF(I)*RF(I)
5835 NEXT I
5840 M=C/U5:V=(B-U5*M*M)/(U5-1):T46=SQR(V)
5845 PRINT"<RMS    ROUGHNESS    COMPLETE> ";T46
5850 REM    **********RAYLINE CLEARANCE
5855 FOR I=1 TO 101:RF(I)=0:NEXT I
5860 RF(1)=D1:RF(U5+1)=D2
5865 FOR I=2 TO U5
5870 RF(I)=I*(D2+Y(U5+1)-D1-Y(1))/U5+D1+Y(1)-Y(I)
5875 NEXT I
5880 PRINT"<RAYLINE CLEARANCE COMPLETE>"
5885 RETURN
```

```
6000 REM   ********************SCALING PLOT
6005 Y1=Y(1):Y2=Y(1)
6010 FOR I=1 TO U5+1
6015 IF(Y1-Y(I))<=0 THEN 6030 ELSE 6020
6020 Y1=Y(I)                                          'MINIMUM Y DATA
6025 GOTO 6040
6030 IF (Y2-Y(I))<0 THEN 6035 ELSE 6040
6035 Y2=Y(I)                                          'MAXIMUM Y DATA
6040 NEXT I
6045 IF Y2<(Y(1)+D1) THEN Y2=Y(1)+D1
6050 IF Y2<(Y(U5+1)+D2) THEN Y2=Y(U5+1)+D2
6055 IF (Y2-Y1)>120 THEN KK=.04 ELSE KK=.2
6060 Y1=INT(KK*(Y1-2.6))/KK
6065 Y2=CINT(KK*(Y2+2.6)+.85)/KK
6070 YD=Y2-Y1
6075 PRINT"<PLOT   SCALING    COMPLETE>"
6080 RETURN
```

```
6200 REM    ********************PLOTTER
6205 IF Y2>850 THEN Z$="#####." ELSE Z$="###.# "
6210 L=YD/U6:WIDTH "LPT1:",100
6215 OPEN "LPT1:" FOR OUTPUT AS #1
6220 W=1:PRINT"<PLOTTING   PATH    PROFILE>"
6225 PRINT #1,"TERRAIN PROFILE  ";N1$;" TO ";N2$
6230 PRINT #1," ":PRINT #1,TAB(10);"HEIGHT (M)";TAB(U6+14);"CLR"
6235 FOR U=0 TO U6/10:PRINT #1,TAB(U*9+8+U-1);:PRINT #1,USING Z$;U*10*L+Y1;
6240 NEXT U:PRINT #1," "
6245 PRINT #1,TAB(10);:FOR I=1 TO U6/5:PRINT #1,"+----";:NEXT I:PRINT #1,"+"
6250 FOR I=1 TO U5+1
6255 GOSUB 4100                                          'WAIT
6260 PRINT #1,MID$("KILOMETERS",W,1);:PRINT #1,TAB(2);:PRINT #1,USING "###.##";X
(I);:
6265 IF I=1 THEN 6310
6270 IF I=U5+1 THEN 6325
6275 PRINT #1,TAB(9)"I";TAB((Y(I)-Y1)/L+10.5);">";TAB(U6+11);"I";
6280 PRINT #1,USING Z$;RF(I)
6285 W=W+1:NEXT I
6290 PRINT #1,TAB(10);:FOR I=1 TO U6/5:PRINT #1,"+----";:NEXT I:PRINT #1,"+"
6295 PRINT #1,USING "          K FACTOR = ##.##";KF;:PRINT #1,USING "    RMS ROUG
HNESS = ####.##";T46
6300 PRINT #1," ":PRINT #1," ":PRINT #1," ":CLOSE #1:WIDTH "LPT1:",80
6305 RETURN
6310 PRINT #1,TAB(9)"I";TAB((Y(I)-Y1)/L+10.5);">";TAB((Y(I)+D1-Y1)/L+11.5);"*";T
AB(U6+11);"I";
6315 PRINT #1,USING Z$;RF(I)
6320 GOTO 6285
6325 PRINT #1,TAB(9)"I";TAB((Y(I)-Y1)/L+10.5);">";TAB((Y(I)+D2-Y1)/L+11.5);"*";T
AB(U6+11);"I";
6330 PRINT #1,USING Z$;RF(I)
6335 GOTO 6285
```

```
7000 REM  **************************PROFILE SCAN FOR ILLUMINATED REGION
7005 Z$="ILLUMINATED PATH"
7010 ALF=SQR(30000/10^(G1/10))/2                    'HALF 3DB BEAMWIDTH
7015 BET=SQR(30000/10^(G2/10))/2                    'HALF 3DB BEAMWIDTH
7020 HA=SIN(P1*8.980001E-03)*(6380.43+Y(1)+D1)
7025 HB=SIN(P1*8.980001E-03)*(6380.43+Y(U5+1)+D2)
7030 HA=ATN(HA/SQR(ABS(1-HA*HA)))*180/3.14159
7035 HB=ATN(HB/SQR(ABS(1-HB*HB)))*180/3.14159
7040 FOR I=1 TO U5+1
7045 THET1=ATN(X(I)/(Y(1)+D1-Y(I)))                 'ANGLE TO GND POINT
7050 IF THET1>HB-BET THEN 7060
7055 X1=I+1
7060 THET2=ATN((P1-X(I))/(Y(U5+1)+D2-Y(I)))         'ANGLE TO GND POINT
7065 IF THET2<HA-ALF THEN 7075
7070 X2=I-1
7075 NEXT I
7080 IF X1>=X2 THEN X1=0:X2=0:Z$="NON EXISTANT"
7085 PRINT"<ANT PROFILE SCAN COMPLETE>"
7090 RETURN

7500 REM  **************************END
7505 CLS:SOUND 88,6:END
```